# Maintainer: Andreas Radke <andyrtr@archlinux.org>
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=libxv
pkgver=1.0.11
pkgrel=4
pkgdesc="X11 Video extension library"
arch=('x86_64')
license=('custom')
url="https://xorg.freedesktop.org/"
depends=('libxext')
makedepends=('xorgproto')
source=(${url}/releases/individual/lib/libXv-${pkgver}.tar.bz2{,.sig})
sha256sums=('d26c13eac99ac4504c532e8e76a1c8e4bd526471eb8a0a4ff2a88db60cb0b088'
            'SKIP')
validpgpkeys=('C41C985FDCF1E5364576638B687393EE37D128F8') # Matthieu Herrb <matthieu.herrb@laas.fr>

build() {
  cd "${srcdir}/libXv-${pkgver}"
  ./configure --prefix=/usr --disable-static 
  make
}

package() {
  cd "${srcdir}/libXv-${pkgver}"
  make DESTDIR="${pkgdir}" install

  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
